
⭐⭐⭐ Challenge ⏱️ 30 minutes

Smart fan

You build a thermostat. Does it get warmer than 25 degrees? Then the relay clicks on. This

way your Arduino can turn on a small fan on a battery, for example. On the screen you see

the temperature and whether the fan is on.

🧰 What do you need?

1× Arduino UNO 1× Temperature sensor

DHT11

1× Relay 5V

1× OLED screen 0.91"

🔌
USB cable + Jumper

wires

🗺️ Wiring diagram

🛠️ This is how you build it

1. Connect the DHT11: DAT to pin 2, VCC to 5V, and GND to GND.

2. Connect the relay: S to pin 7, + to 5V, and − to GND.

3. Connect the OLED screen: SDA to A4, SCL to A5, VCC to 5V, and GND to

GND.

4. Do you really want to switch something? Then only use a battery or low

voltage on COM and NO, never the wall outlet.

💻 The code

🧠 What does the code do?

The DHT11 measures the temperature. Above 25 degrees we turn the

relay on.

Only below 24 degrees does the relay turn off again. With those two

limits, it does not keep clicking on and off.

The OLED screen shows the temperature and "Fan: ON" or "off".

🎯 Try this…

Add two buttons to set the limit temperature higher and lower

Turn on an LED while the fan is on

Let the fan also turn on when the humidity goes above 70%

🚀 Build this project with Codey: open codey.online/starter-kits/uno/projects/thermostat and click MAKE THIS. Then the code and the wiring diagram

are ready right away.

Arduino UNO

Temperature sensor DHT11

Relay 5V

OLED screen 0.91"

// Smart fan (thermostat)

// Is it getting too warm? Then the relay turns on, for example for a fan.

// On the screen you see the temperature and whether the fan is on.

#include <Wire.h>

#include <Adafruit_GFX.h>

#include <Adafruit_SSD1306.h>

#include <DHT.h>

const int DHT_PIN = 2;        // DAT of the DHT11 on pin 2

const int RELAY_PIN = 7;      // S of the relay on pin 7

const float ON_ABOVE = 25.0;  // Above this temperature the fan turns on

const float OFF_BELOW = 24.0; // Below this temperature it turns off again

DHT dht(DHT_PIN, DHT11);

Adafruit_SSD1306 display(128, 32, &Wire, -1);  // OLED: SDA = A4, SCL = A5

bool fanOn = false;

void setup() {

  Serial.begin(9600);

  pinMode(RELAY_PIN, OUTPUT);

  digitalWrite(RELAY_PIN, LOW);   // Start with the relay off

  dht.begin();

  display.begin(SSD1306_SWITCHCAPVCC, 0x3C);

  display.setTextColor(SSD1306_WHITE);

}

void loop() {

  float temperature = dht.readTemperature();

  if (!isnan(temperature)) {

    // Two limits, so the relay does not keep clicking on and off

    if (temperature > ON_ABOVE) fanOn = true;

    if (temperature < OFF_BELOW) fanOn = false;

    digitalWrite(RELAY_PIN, fanOn ? HIGH : LOW);

  }

  display.clearDisplay();

  display.setTextSize(2);

  display.setCursor(0, 0);

  if (isnan(temperature)) {

    display.print(F("DHT11 ?"));

  } else {

    display.print(temperature, 1);

    display.print((char)247);

    display.print(F("C"));

  }

  display.setTextSize(1);

  display.setCursor(0, 22);

  display.print(fanOn ? F("Fan: ON") : F("Fan: off"));

  display.display();

  Serial.print(F("Temperature: "));

  Serial.print(temperature);

  Serial.println(fanOn ? F(" C  -> ON") : F(" C  -> off"));

  delay(2000);

}

http://app/starter-kits/uno/tutorial/kit/dht11-module
http://app/starter-kits/uno/tutorial/kit/relay-module-5v
http://app/starter-kits/uno/tutorial/kit/oled-display-091-inch

